BAYESIAN APPROACHES TO MIXED EFFECTS MODELS FOR ORDERED CATEGORICAL DATA

metrum institute

W.R. Gillespie, PhD, M.R. Gastonguay, PhD, W. Knebel, PhD, Metrum Institute, Tariffville, CT 06081

ABSTRACT

Purpose: To analyze ordered categorical data via Bayesian methods and compare performance to that of NONMEM with respect to prediction and parameter estimation.

Methods: Hypothetical ordered categorical data from dose-response trials were simulated using cumulative logit models. Each trial was a parallel design with 4 dose arms (0, 7.5, 15, 30), 250 patients/arm and 4 observations/patient (incl. baseline). Each datum was a 4 level ordered categorical score (0, 1, 2, or 3). Three cases were simulated (100 replicates each): (1) non-skewed, low inter-individual variance (IIV); (2) skewed, low IIV; (3) skewed, high IIV. Bayesian and NONMEM implementations of a cumulative logit model were used to analyze the data. Bayesian analyses used Markov chain Monte Carlo (MCMC) simulation implemented in OpenBUGS. NONMEM used the Laplacian approximation.

Results: The results confirmed previous reports of bias and imprecision in NONMEM parameter estimates that increased with increasing IIV and skewness. MCMC estimated posterior means showed minimal bias for all 3 cases. When IIV was large, biased NONMEM parameter estimates caused overestimation of rare event rates. MCMC estimated posterior expected rates showed minimal bias.

Conclusions: Bayesian analysis of repeated ordered categorical data using MCMC results in more accurate and precise parameter estimates and predictions than the NONMEM Laplacian method when the true model is skewed and IIV is large. Other advantages and disadvantages of Bayesian modeling are discussed. A major disadvantage is the lack of Bayesian modeling software with a library of built-in PK & PD models. An effort is proposed to address that limitation.





PURPOSE

- To implement cumulative logit models for ordered categorical data using OpenBUGS plus R tools.
- To compare performance of OpenBUGS and NONMEM with respect to prediction and parameter estimation.
- To discuss limitations of available Bayesian modeling tools for clinical pharmacology applications.
- To propose efforts to address some of those limitations

CONTEXT & MOTIVATION

Kjellsson MC,[1] Jönsson S,[1] and Karlsson MO.[1] **The Back-Step Method—Method for Obtaining Unbiased Population Parameter Estimates for Ordered Categorical Data.** AAPS J. 2004 Aug 11;6(3):e19.

Jönsson S,[1] Kjellsson MC,[1] and Karlsson MO.[1] **Estimating Bias in Population Parameters for Some Models for Repeated Measures Ordinal Data using NONMEM and NLMIXED.** J Pharmacokinet Pharmacodyn. 2004 Aug;31(4):299-320.

[1]Division of Pharmacokinetics and Drug Therapy, Dept. of Pharmaceutical Biosciences, Uppsala University, Box 591, SE-751 24 Uppsala, Sweden

Kjellsson, Jönsson & Karlsson simulation exercises:

- Ordered categorical responses (4 levels)
- NONMEM Laplacian method results in estimation and prediction biases
  - Particularly when the data are skewed to one extreme and/or inter-individual variation (IIV) is large
  - Probabilities of rare events are overestimated
- Illustrated 2 approaches for reducing that bias:
  - The back step method, an iterative application of NONMEM
  - A Gaussian quadrature method (NLMIXED in SAS)

CONTEXT & MOTIVATION (cont.)

Bayesian modeling using Markov chain Monte Carlo (MCMC) simulation

- Provides results in the form of samples from the joint posterior distribution of the model parameters
- Should not produce the same biases as the Laplacian approximation
- The work presented here tests that expectation by applying MCMC to the same simulated cases as Kjellson et al

METHODS

- Trial simulations performed using R
- Same model & parameter values as Kjellsson et al
- Trial design:
  - 4 dose arms: 0, 7.5, 15, 30
  - 250 patients per arm
  - 4 observations per patient (baseline + 3)
- 100 trial replicates per scenario

Model used for simulation and analysis

The score (0, 1, 2 or 3) at the ith occasion in the jth individual (Yij) is described by:

$$\text{logit}\left(\Pr\left(Y_{ij} \geq m|\theta,\omega\right)\right) \sim N\left(\mu_{m,ij},\omega^2\right)$$

$$\mu_{m,ij} = \sum_{k=1}^{m}\theta_k + I_{t_{ij}>0}\left(\theta_4 + \theta_5 D_j\right)$$

Case	Parameter values					
	θ_1	θ_2	θ_3	θ_4	θ_5	ω^2
1	1.85	-1.85	-1.85	0.483	0.046	4
2	-4.88	-0.548	-1.18	1.55	0.03	4
3	-11.8	-1.32	-2.96	3.85	0.717	40

Case	Expected fraction of baseline scores			
	0	1	2	3
1	0.24	0.26	0.26	0.24
2	0.965	0.0122	0.0144	0.0084
3	0.965	0.0122	0.0144	0.0084

OpenBUGS implementation

- Simulated trials analyzed using OpenBUGS + BRugs (R interface to OpenBUGS)
  - Model identical to that used for simulation except for presence of prior distributions
  - Relatively uninformative priors
  - MCMC settings:
    - 3 chains
    - Burn-in for 4001 samples/chain
    - 5010 post-burn-in samples/chain (keep every 15th)

NONMEM Model

```
$PRED

; indicator for post-baseline data
 IPOST = 0
 IF (TIME .GT. 0) IPOST = 1

; treatment effect
 ETREAT = IPOST*(THETA(4) + THETA(5)*DOSE)

; logits for cumulative probabilities
 LPCUM1 = THETA(1) + ETREAT + EXP(THETA(6))*ETA(1) ; SCORE >= 1
 LPCUM2 = LPCUM1 - EXP(THETA(2))                   ; SCORE >= 2
 LPCUM3 = LPCUM2 - EXP(THETA(3))                   ; SCORE >= 3

; cumulative probabilities
 PCUM1 = (1/(1+EXP(-LPCUM1)))
 PCUM2 = (1/(1+EXP(-LPCUM2)))
 PCUM3 = (1/(1+EXP(-LPCUM3)))

; probabilities for each score (likelihood)
 P0 = 1 - PCUM1
 P1 = PCUM1 - PCUM2
 P2 = PCUM2 - PCUM3
 P3 = PCUM3

; indicators for each score
 I0=0
 I1=0
 I2=0
 I3=0
 IF (DV.EQ.0) I0=1
 IF (DV.EQ.1) I1=1
 IF (DV.EQ.2) I2=1
 IF (DV.EQ.3) I3=1

; likelihood
 Y = P0*I0 + P1*I1 + P2*I2 + P3*I3

$ESTIMATION MAX=9999 PRINT=1 METHOD=COND LAPLACE LIKE NOABORT
```

BUGS model

```
model{

    for(i in 1:npat){

        # interpatient variability
        eta[i] ~ dnorm(0,tau.eta)

    }

    for(i in 1:nobs){

        # likelihood for observed score
        score[i] ~ dcat(p[i,1:4])

        # probabilities for each score
        p[i,1] <- 1 - pcum[i,1]
        p[i,2] <- pcum[i,1] - pcum[i,2]
        p[i,3] <- pcum[i,2] - pcum[i,3]
        p[i,4] <- pcum[i,3]

        # treatment effect model & calculation of cumulative
        # probabilities
        logit(pcum[i,1]) <- theta[1] + (theta[4] +
              theta[5]*dose[i])*(1-equals(time[i],0)) +
              eta[patient[i]]
        logit(pcum[i,2]) <- logit(pcum[i,1]) + theta[2]
        logit(pcum[i,3]) <- logit(pcum[i,2]) + theta[3]

    }

    # prior distributions
    theta[1] ~ dnorm(0,0.00001)
    theta[2] ~ dnorm(-1,0.00001)I(,0)
    theta[3] ~ dnorm(-1,0.00001)I(,0)
    theta[4] ~ dnorm(0,0.00001)
    theta[5] ~ dnorm(0,0.00001)
    sigma.eta ~ dunif(0,1000)
    tau.eta <- 1/(sigma.eta*sigma.eta)
}
```

RESULTS

Relative bias in parameter estimates



- Bias in NONMEM parameter estimates increases with increasing IIV and skewness
- MCMC estimated posterior means show minimal bias for all 3 cases

Relative root mean square error



- MCMC estimated posterior means consistently result in RMSE $\leq$ that for NONMEM estimates

Predicted fractions of responses by score



- When IIV is large biased NONMEM parameter estimates cause overestimation of rare event rates.
- MCMC estimated posterior expected rates show minimal bias

Copies of this poster are available at: www.metruminstitute.org/publications/index.shtml

DISCUSSION/CONCLUSIONS

The case for Bayesian modeling of ordinal data using MCMC

- Better estimation and prediction performance than methods using linear or Laplacian approximation to the likelihood
- Yields an estimate of the entire joint posterior distribution of the model parameters
  - Describes uncertainty in parameters
  - Uncertainty in derived quantities, e.g., predictions, is easily calculated from MCMC samples
- Can easily and rigorously include prior information
- Available tools, e.g., WinBUGS/OpenBUGS, permit very flexible model specification:
  - Rich collection of built-in probability distributions
  - No limit on levels of variability

The case against Bayesian modeling of ordinal data using MCMC

- Requires more computation time
  - ~ 15-45 minutes per trial (elapsed time with Intel Core Duo 2.33 GHz, 2 GB RAM)
  - Limited benefit from parallel computation
    - Though substantial gains are possible by running multiple chains in parallel
  - NONMEM requires substantially less time to obtain point estimates
  - SAS NLMIXED using Gaussian quadrature is also faster
    - But if you want rigorous characterization of uncertainty with ML methods:
      - Bootstrapping is probably the best option
      - And that also requires sizable computation time
      - But it is readily accelerated via parallel computation

Primary limitation of existing Bayesian software for clinical pharmacology applications

- Lack of built-in PK & PD models
  - PKBugs limited to linear compartmental models with · 3 compartments
    - Also not available for OpenBUGS or latest version of WinBUGS
  - Support for ODEs available (WBDIFF & MCSim) but substantial programming needed to apply it to multiple dose data
  - In short, no real equivalent to PREDPP

Metrum Institute efforts to address limitations & facilitate use of Bayesian methods in pharmacometrics

- Develop computational methods and open-source software tools for Bayesian modeling and simulation relevant to pharmacometric applications
  - Short term efforts
    - Currently developing a PREDPP equivalent for WinBUGS and OpenBUGS
    - Develop tools for distributed computing of multiple chains
      - Working prototype using MPICH2 and R with a modified version of R2WinBUGS
  - Tentative long-term plans:
    - Develop a more comprehensive platform for Bayesian M&S
      - Implement both MCMC and estimation of posterior modes.
      - Efficient estimation of posterior modes would facilitate rapid exploratory modeling
      - Open source with greater platform (s/w & h/w) independence
      - Support for parallel computing
      - Suite of tools for analysis of MCMC samples
      - Probably structured as one or more R packages
- Provide short courses in Bayesian modeling for pharmacometric applications